CIVIL ENGINEERS
POCKET COMPUTER MONTHLY

GENERAL COMPUTATION PROGRAM

Introducing GCP — a New Program for Engineers — A Formula Solver

Requires only a limited degree of computer programming skill.

Problems are solved by the engineer in the traditional way:

1) Write the formulas;
2) Enter the values of input parameters;
3) The computer calculates the answers.

GCP requires that the formulas be written in BASIC code beginning with Program Line 100.

Four versions of GCP are listed herewith:

A) PC-2 with printer
B) PC-2 without printer
C) PC-1 with printer
D) PC-1 without printer

Keep GCP in the computer's memory so that the solutions to problems involving formulas may be quickly obtained. Before entry of new formulas, be sure the old formulas have been erased from memory. (To erase a program line, enter the line number while in the PRO mode, then press "Enter.") Use program lines 100-1200 in PC-2 (100-500 in PC-1).

After your formulas have been entered, you may, at your option, preserve the program on a cassette. In this way you can build up a library of customized GCP's. Of course, you may prefer to write the formulas into memory in PRO mode, each time the problem arises.

Example #1: SIMPLE RECTANGULAR BEAM

The Formula for Moment in a uniformly loaded simple beam is $1/8\ wl^2$.

For a rectangular beam, the Section Modulus equals $1/6\ bd^2$.

The flexural stress, f_b, equals the Moment divided by the Section Modulus, M/S.

The shear stress, f_v, equals 3/2 · V/A
where V = wl/2
and A = bd

The input parameters are:
b (beam width, inches)
d (beam depth, inches)
w (load per foot)
l (span, feet)

These input parameters are assigned alphabetical designations, beginning with "I":

I = b = beam width, inches
J = d = beam depth, inches
K = w = load, lb/ft.
L = l = span, ft.

Output parameters are then assigned alphabetical designations beginning with "A". The formulas are rewritten in BASIC Code: (This should be quite easy for even those engineers who have never studied BASIC programming.)

$M = 1/8\ wl^2$
becomes `100 A = K/8 * L^2`

$S = 1/6\ bd^2$
becomes `110 B = I/6 * J^2`

f_b= 12 M/S
becomes `120 C = 12 * A/B`

f_v= 3/2 wl/(2 bd)
becomes `130 D = 3/2 * K * L/2/I/J`
or, alternatively
`130 D = 3/2 * K * L/(2*I*J)`

With the computer in the PRO mode, simply enter the line number (e.g. 100), followed by the formula, into the program. After each formula has been typed, press

CIVIL ENGINEERS POCKET COMPUTER MONTHLY

"Enter". To check the formulas for correctness (PC-2), Type LL. 100, 130 and press "Enter" which will result in a printout of the formulas. On PC-1, you may type L. and press "Enter", and the entire program will be listed. The computer screen may of course be used to check for correctness.

To run the program press DEF A (Shift A on PC-1). Watch the display and follow the instructions.

To rerun the program, press DEF C (Shift C on PC-1). The rerun feature allows fast entry of new parameter values, with a minimum of wasted effort.

The answers are printed out in this order: A thru H. Thus, up to eight parameter values may be output.

NOTE: In Versions A and B of GCP (PC-2) you may, at your option, label the input and output parameters.

Example #2: PILE DRIVING INSPECTION

Determine the bearing value of the pile using the following dynamic formula for mechanical hammers:

$$R = \frac{2.5\ E}{S+0.1} C$$

Where:

$$C = \frac{W_r + e^2 W_p}{W_r + W_p}$$

R = estimate of unit pile capacity, tons

S = the "set", i.e., the average penetration in inches per blow, for the last ten blows of the hammer, or alternatively, for the last foot of driving

E = driving energy, ft.-tons = $W_r \cdot h$ for single-acting hammers

W_r = weight of ram, tons

h = height of fall, feet, single-acting hammers

W_p = Est. Weight of the pile, tons

e = coefficient of restitution of impact area. Use:

- 0.65 phenolic laminate
- 0.55 steel on steel
- 0.5 hardwood blocks
- 0.25 softwood blocks
- 0 excessively soft packing or deteriorated blocks

The parameters S, W_p, W_r, e, E are entered as I, J, K, L and M. Note that the value of set, S, is not equal to I, but rather is a function of I. See below.

After the initial computation, only the values, I and J (weight of pile), need be entered to accomplish succeeding computations. (Enter I, press "Enter," then J, then press "Enter" twice. PC-2 only)

The formulas:

```
100 A = 12/I
      where
      I = blows per foot, last foot (If the
          number of inches of penetra-
          tion are measured for the
          last 10 blows, the formula
          becomes A = I/10)
110 B = (K + L^2 * J)/(K + J)
120 C = 2.5 * M * B/(A + 0.1)
```

Alternatively, if only the pile capacity is required, the last formula is assigned to A, thus:

```
120 A = 2.5 * M * B/(A + 0.1)
```

and the value, A, (pile capacity, tons) is the only output parameter.

Example #3: COLUMN DESIGN - MAIN MEMBERS

Ref: AISC "Manual of Steel Construction" (8th Ed., 1980). Design of a W, Tube or Pipe Column for Axial Load. Program solves:

1) Rolled, or welded plate (built-up), H or I-shaped Columns, designated "W".
2) Pipe Columns, designated "P".
3) Tube (rectangular or square) Columns, designated "T".

Due to lack of memory capacity, PC-1 requires three programs so as to solve W, P, and T columns, respectively. We find that GCP in PC-1 will handle only about 10-15 program lines. Since the unaltered PC-2 has about 2K RAM, any GCP program which will work in PC-1 will, of course, also work in any PC-2. (Our PC-2 has the 8K RAM module.)

This is a trial-and-error design process. Therefore, set up a table of several trial sections, listing I_x, I_y, A, d, b_f, t_f, F_y.

The program:

1) Calculates $K_x I_x/r_x$ and $K_y I_y/r_y$; determines which is critical (larger).

2) If $KI/r \geq 200$, the program prints a message and stops.

3) For W columns, program checks flange width and, if necessary, calculates a reduction factor, Q_s where $Q_s \leq 1$. See pp. 5-29, 5-30, and 5-94, AISC.

4) For Tube columns, program calculates ratio of width to thickness and, if necessary, calculates b_e (effective width) Using b_e, it calculates a reduction factor, Q_a, where $Q_a \leq 1$. See pp. 5-30 and 5-95, AISC.

5) For Pipe Columns, program checks thickness and, if necessary, calculates F_a(allowable axial stress in pipe where limited by diameter/ thickness ratio). If pipe wall is too thin (diam/ thickness too great) the program prints a message and stops.

6) Calculates C_c and the allowable axial stress, F_a, using either the CRC or Euler formula as appropriate. (Uses the Euler formula if KI/r exceeds C_c). See pg. 5-19 AISC. Formulas are modified by safety factor as discussed below. Note that F_a is modified by Q_a and Q_s as applicable. See Page 5-96.

Input Parameter Designations:

I = I_x (Mom. of Inertia, x, $in.^4$)
J = I_y (Mom. of Inertia, y, $in.^4$)
K = A (Cross-sect. Area, $in.^2$)
L = d (Depth of Member, in.), W section
M = b_f (Flange Width, in.), W section
N = t_f (Flange Thickness, in.), W section
O = F_y (Specified Min. Yield Stress, ksi)
P = P (Applied Axial Load, kips)
Q = K_x(Effective Length Factor, x Direction)
R = K_y(Effective Length Factor, y Direction)
S = L_x(Length, Unbraced, x direction, Ft.)
T = L_y(Length, Unbraced, y Direction, Ft.)

For Tubes: L = depth of tube, in.
M = width of tube, in.
N = metal thickness, in.

For Pipes: L = M = Outside Diam., in.
N = metal thickness, in.

To determine values of K_x and K_y, see AISC Spec. para. 1.8, pp. 5-29 and 5-123 to 127; Table C-1.8.1, Pg. 5-124. Note that for braced frames, where the top and bottom of the column are maintained in alignment, Cases (a) or (b) of Table C-1.8.1 apply; however, for a pin-ended (Case d) column, K=1. We believe that designers should tend toward the use of K=1 because of the possibility of inevitable imperfections of straightness and shape, because of out-of-plumbness or other accidental eccentricities of loading (non-axiality), and because of locked-in stresses within the metal associated, for example, with differential rates of cooling.

In cases where the frame is unbraced (sidesway not prevented), Cases (c), (e), and (f) of Table C-1.8.1 apply, and K > 1.

For continuous frames, see Alignment Chart, AISC pp. 5-125; however, for these cases, combined axial and bending loading will no doubt, be applicable. See February issue!

The primary output parameters are 1) A = axial stress, f_a, and 2) B = allowable axial stress, F_a.

This program will be particularly useful for non-typical design situations, such as built-up columns, non-standard shapes, and where F_y is not equal to 36 or 50 ksi.

Type these lines into the program, press DEF A (Shift A on PC-1) and follow the instructions in the display.

Line 100 allows user to designate which type of column cross-section (W, P, or T) is to be designed. Line 110-130 calculates two values of kl/r, selects larger value, and determines whether kl/r exceeds 200. If so, program stops.

PC-2

```
100:INPUT "ENTER(W
    ),(T)UBE,OR (P
    )IPE";U$:V=1:W
    =1:X=0
110:A=Q*S*12/(I/K)
    ^.5:B=R*T*12/(
    J/K)^.5
120:C=A:IF B>CLET
    C=B
130:IF C>200PRINT
    "KL/R>200:INVA
    LID":END
```

Lines 140-180 are for W columns. Checks unstiffened flange-thickness ratio (½ of the flange width, divided by flange thickness) to determine whether a stress reduction factor, V (Q_s), is required. If so, calculates V.

PC-2

```
140:IF U$<>"W"THEN
    200
150:IF M/2/N<=95/(
    O)^.5THEN 200
160:IF M/2/N>176/(
    O)^.5THEN 180
170:V=1.415-.00437
    *M/2/N*O^.5:
    GOTO 200
180:V=20000/(O*(M/
    2/N)^2)
```

Lines 200-240 are for tube columns, square or rectangular. Calculates Z (b_e) and if Z is less than the maximum tube width, calculates stress reduction factor, W (Q_a).

PC-2

```
200:IF U$<>"T"THEN
    300
210:Y=L:H=M:IF M>Y
    LET Y=M:H=L
220:IF Y/N<=238/O^
    .5THEN 300
230:Z=253*N/(P/K)^
    .5*(1-50.3/((Y
    /N)*(P/K)^.5))
    :IF Z>YLET Z=Y
250:W=(Z+H)/(Y+H)
```

Lines 300-330 are for pipe columns. Calculates diam/thickness ratio (L/N) and if too large, calculates maximum working stress, F_a; it will stop the program if ratio is larger than 13000/F_y.

PC-2

```
300:IF U$<>"P"THEN
    ~~400~~ 380
310:IF L/N<=3300/O
    THEN 400
320:IF L/N>13000/O
    PRINT "INVALID
     COLUMN":END
330:D=662/L*N+.4*O
    :X=1
```

Lines 400-410 calculate C_c (E); then if kl/r (C) is greater than C_c, the Euler formula is used to calculate F_a (F), modified by safety factor 23/12. If not, the CRC formula is used, modified by a variable safety factor. Line 420 accounts for the case where a value of F_a (D) was calculated for a pipe column, in line 320.

PC-2

```
380:IF V>1LET V=1
390:IF W>1LET W=1
400:E=(58E3*π*π/(V
    *W*O))^.5:IF C
    >ELET F=12*π*π
    *29E3/(23*C*C)
    :GOTO 420
410:F=(1-C*C/2/E/E
    )*O:F=F/(5/3+3
    /8*C/E-C*C*C/8
    /E^3)
420:IF X=1AND F>D
    LET F=D
430:X=E:E=C:B=F*V*
    W:A=P/K:D=F*K*
    V*W:C=P:F=X
450:IF B<APRINT "C
    OL.OVERSTRESSE
    D"
```

CIVIL ENGINEERS POCKET COMPUTER MONTHLY

GCP Listings (PC-1)

W Columns

```
110:A=QS*12/(I/K
    )^.5:B=RT*12
    /(J/K)^.5:V=
    1
120:C=A:IF B>C
    LET C=B
130:IF C>200
    PRINT "KL/R>
    200:INVALID"
    :PRINT " ":
    PRINT " ":
    GOTO 555
150:IF M/2/N<=95
    /O^.5THEN 38
    0
160:IF M/2/N>176
    /O^.5THEN 18
    0
170:V=1.415-.004
    37M/2/N*O^.5
    :GOTO 380
180:V=20000/(O*(
    M/2/N)^2)
380:IF V>1LET V=
    1
400:E=(2π^2*29E3
    /(VO))^.5:IF
    C>ELET F=12π
    ^2*29E3/(23*
    C^2):GOTO 42
    0
410:F=(1-CC/2/E/
    E)*O:F=F/(5/
    3+3/8*C/E-CC
    C/8/E^3)
420:X=E:E=C:B=FV
    :A=P/K:D=FKV
    :C=P:F=X
```

Tube Columns

```
110:A=12QS/(I/K)
    ^.5:B=12RT/(
    J/K)^.5:W=1
120:C=A:IF B>C
    LET C=B
130:IF C>200
    PRINT "KL/R>
    200:INVALID"
    :PRINT " ":
    PRINT " ":
    GOTO 555
150:Y=L:H=M:IF M
    >YLET Y=M:H=
    L
160:IF Y/N<=238/
    O^.5THEN 380
170:Z=253N/(P/K)
    ^.5*(1-50.3/
    ((Y/N)*(P/K)
    ^.5)):IF Z>Y
    LET Z=Y
180:W=(Z+H)/(Y+H
    )
380:IF W>1LET W=
    1
400:E=(π*π*58E3/
    (WO))^.5:IF
    C>ELET F=12π
    *π*29E3/(23C
    C):GOTO 420
410:F=(1-CC/2/E/
    E)*O:F=F/(5/
    3+3/8*C/E-CC
    C/8/E^3)
420:X=E:E=C:B=FW
    :A=P/K:D=FKW
    :C=P:F=X
```

Pipe Columns

```
110:A=QS*12/(I/K
    )^.5:B=RT*12
    /(J/K)^.5:X=
    0
120:C=A:IF B>C
    LET C=B
130:IF C>200
    PRINT "KL/R>
    200:INVALID"
    :PRINT " ":
    PRINT " ":
    GOTO 555
150:IF L/N<=3300
    /OTHEN 400
160:IF L/N>13000
    /OPRINT "INV
    ALID COLUMN"
    :GOTO 555
170:D=662/L*N+.4
    *O:X=1
400:E=(2π^2*29E3
    /O)^.5:IF C>
    ELET F=12π^2
    *29E3/(23*C^
    2):GOTO 415
410:F=(1-CC/2/E/
    E)*O:F=F/(5/
    3+3/8*C/E-CC
    C/8/E^3)
415:IF X=1IF F>D
    LET F=D
420:X=E:E=C:B=F:
    A=P/K:D=FK:C
    =P:F=X
```

GRAPHICS

The enclosed listing permits plotting of the CRC-Euler curve; a unique curve is plotted for each yield stress.

The length of the ordinate scale is appropriate for F_y values up to about 65 ksi; this could of course be modified within the program for greater values of F_y.

This curve represents the estimated buckling failure stress, versus kl/r. It is apparent from a study of this curve why kl/r values greater than 200 are not deemed prudent.

For main members with kl/r values greater than C_c, the required safety factor is 23/12; for members with kl/r equal to C_c or less, the CRC curve is used, modified by a safety factor varying from 5/3 at kl/r=Ø, to 23/12 at kl/r=C_c.



COLUMN CURVE

This compound curve is based on experience and actual performance of steel columns during testing and research. Obviously, steel cannot be stressed at a higher level than F_y (yield stress). Therefore the curve intersects the ordinate at F_y. It was found that the CRC curve approximates the experimental data; the Euler Curve is valid at kl/r values higher than C_c (C_c equals kl/r at $F_y/2$).

CIVIL ENGINEERS POCKET COMPUTER MONTHLY

NEWS

Sharp apparently intends to replace PC 1211 (Radio-Shack's PC-1) with their new PC-1250, an upgrade of PC-1211. Enhanced BASIC, better handling of strings, and two-dimensional arrays will be among the features of this improved pocket unit. Possibly faster—not sure. Price? Probably about $110. A thermal printer peripheral, with micro cassette recorder built in, will sell for about $170. Our comment: Sounds good, but a larger RAM than 2K sure would have been nice!

PROGRAMMERS TIP

Use your PC-2 as an electric typewriter. Types with automatic return. Variable type size (Csize 1-4). However, watching it type "vertically" takes some getting used to!

Note the use of INKEY to assign letters, numbers, etc. to A$. The computer "waits" in line 9020 until something (letter, number, etc.) is typed. Also note the use of ASC in lines 9030 and 9040 to convert A$ to a number for purposes of line shifting, so that you may use the "up" and "down" arrows (ASC numbers 10 and 11, resp.) to change lines. Line 9110 shifts up, and line 9120 shifts down. Use the arrow key on the printer to start a new "page". For rapid repeats, hold the key down.

Suggestion: add this short routine to the end of certain programs to permit custom-typing of a message. To activate, press DEF=. When done, press BREAK (ON).

```
9000: "="GRAPH : INPUT "CSIZE ?(1-4)";
      A:CSIZE A:ROTATE 1:GLCURSOR
      (190,0):SORGN
9005: B=INT (80/A+.4)
9010: FOR X=0 TO B
9020: V$=INKEY$:IF V$="" THEN 9020
9030: IF ASC V$=10 THEN 9110
9040: IF ASC V$=11 THEN 9120
9050: LPRINT V$:V$="":NEXT X
9110: GLCURSOR (-10*A,0):SORGN:
      GOTO 9010
9120: GLCURSOR (10*A,0):SORGN:GOTO
      9010
9999: END
```

PC-1 Owners: If you have bought Radio Shack's "Civil Engineering" software, take note of an error in the Section Properties program ("SECT") as follows:

Line 199: A=D-C:E=-E:GOTO 185
should be: A=D-C:E=-E:C=-C:GOTO 185
which corrects the calculation of Radius of Gyration of Symmetric Box, I, or C Section.

GENERAL NOTES

We would enjoy hearing from you.We want our efforts to be responsive to your needs in Civil Engineering practice. Our purpose, as we see it, is to help you become more efficient in your engineering work through the use of the computer. To us, the computer is a "means to an end", the "end" being more efficiency and more accuracy - a higher plane of engineering practice. We think the portable units have a distinct advantage over the non-portable. If you have a contribution to make, for example, a program to submit, please do so. No mailings returned. Cliff Hall, Editor, CECOM, P.O. Box 6, Hart, MI 49420.

NEXT ISSUE

The February Issue will present the complete AISC analysis procedure GCP programming for Combined Axial and Bending of Steel Members using PC-2. For PC-1/PC-2 users we will use GCP to solve wood foundation walls using National Forest Products Association (NFPA) procedure.

CASSETTES

We offer cassette tapes with documentation at $20 each for:

1) Statistics-Confidence Program PC-1
2) Statistics-Confidence Program PC-2
3) GCP with or without Printer PC-1
4) GCP with or without Printer PC-2
5) Accounts Receivable PC-2*

*This program prints detailed invoices. Prepares summary of business during the time period. Suitable for a small engineering office (up to five hourly categories of charge rates).

Listing - GRAPHICS COLUMN CURVE

```
  3:"A"CLEAR :
    GRAPH :
    GLCURSOR (0,-2
    00)
  4:INPUT "Fy,KSI=
    ";FY:SORGN :
    CSIZE 1
  6:LINE -(0,260):
    LINE (0,0)-(20
    0,0)
  7:FOR I=80TO 240
    STEP 80:
    GLCURSOR (0,I)
    :RLINE (0,0)-(
    5,0):LPRINT I/
    4:NEXT I
  8:FOR I=40TO 200
    STEP 40:
    GLCURSOR (I,0)
    :RLINE (0,0)-(
    0,4)
  9:GLCURSOR (I-9,
    -10):LPRINT 2*
    I:NEXT I
 10:L1=0:F1=FY
 13:CSIZE 2
 14:CC=π*(2*29E3/F
    Y)^.5
 15:GLCURSOR (3,26
    0):LPRINT " Fa
    ,KSI":GLCURSOR
    (150,-30):
    LPRINT "(kl/r)
    ":GLCURSOR (CC
    /2,0):RLINE (0
    ,0)-(0,6)
 16:GLCURSOR (CC/2
    -5,10):LPRINT
    "Cc":GLCURSOR
    (CC/2-10,FY*2)
    :RLINE (0,0)-(
    20,0):GLCURSOR
    (CC/2+10,FY*3)
 17:LPRINT "Fy=";F
    Y;"KSI":
    GLCURSOR (CC/2
    +10,FY*3-18):
    LPRINT "Cc=";
    INT (CC*10+.5)
    /10
 19:GLCURSOR (CC/4
    ,3.7*FY):
    LPRINT "CRC Cu
    rve":GLCURSOR
    (.65*CC,25):
    LPRINT "Euler
    Curve"
 20:GOTO 35
 30:L1=L2
 35:L2=L1+2.5:IF L
    2>400THEN 80
 37:IF ZZ=1THEN 50
 40:F2=FY*(1-.5*(L
    1/CC)^2):IF F2
    <=FY/2LET ZZ=1
    :GOTO 50
 45:GOTO 60
 50:F2=π^2*29E3/L2
    ^2
 60:LINE (L1/2,F1*
    4)-(L2/2,F2*4)
 70:F1=F2:GOTO 30
 80:TEXT :LF 3:
    GRAPH :CSIZE 3
    :LPRINT "COLUM
    N CURVE":TEXT
    :CSIZE 2:LF 6
999:END
```

Listing - GCP (PC-1 with printer)

```
 10:"A"CLEAR
 20:BEEP 4:PAUSE
    "*****GCP***
    **":PAUSE "I
    NPUT PARAMS:
    A-B" I-T "
 25:BEEP 2:INPUT
    "# INPUT PAR
    AMS?(12MAX)"
    ;A(53)
 30:I$="I":J$="J
    ":K$="K"
 31:L$="L":M$="M
    ":N$="N":O$=
    "O":P$="P":Q
    $="Q":R$="R"
    :S$="S":T$="
    T"
 35:FOR A=9TO 8+
    A(53)
 40:A$(A+26)=A$(
    A):NEXT A
 67:A(54)=0:GOTO
    70
 69:"C"PAUSE "GC
    P-RERUN":A(5
    4)=1:FOR A=9
    TO 8+A(53):A
    $(A)=A$(A+26
    ):NEXT A
 70:BEEP 2:FOR A
    =9TO A(53)+8
    :PAUSE "INPU
    T":PAUSE A$(
    A);"..";A$(A
    );"..";A$(A)
    ;"..";A$(A):
    INPUT A(A):
    NEXT A
 75:IF A(54)=1
    THEN 90
 80:BEEP 2:PAUSE
    "AVAIL:8 OUT
    PUT PARA:A-H
    ":BEEP 2:
    INPUT "NO.OF
     OUTPUT PARA
    MS?";A(55)
 90:PAUSE "CALCU
    LATION"
501:BEEP 2:PRINT
    "  GEN COM P
    GM":PRINT "
       RESULTS"
503:FOR A(56)=1
    TO A(55):A(A
    (56)+26)=A(A
    (56)):NEXT A
    (56)
504:A$="A=":B$="
    B=":C$="C=":
    D$="D=":E$="
    E=":F$="F="
507:G$="G=":H$="
    H="
510:FOR A(56)=1
    TO A(55):
    PRINT A$(A(5
    6))
520:PRINT A(A(56
    )+26)
530:NEXT A(56):
    PRINT " "
540:FOR A(56)=9
    TO A(53)+8:
    PRINT A(A(56
    )):NEXT A(56
    )
550:PRINT " ":
    PRINT " ":
    PRINT " "
555:INPUT "RERUN
    ?PRESS SHFT
    C";Z
560:END
```

Listing - GCP (PC-2 with printer)

```
8:"A"WAIT 70:
  CLEAR
9:PRINT "GENERAL
   COMPUTATION P
  GM"
10:REM
11:REM    PROGRAM
   INSTRUCTIONAL
   MESSAGE
12:REM
13:LS$="**    **
  **    **"
15:IM$="N":INPUT
  "INSTRUCTION M
  SG?(Y/N)";IM$
16:IF IM$="N"THEN
  25
17:PRINT "BEFORE
  RUNNING, BE SU
  RE"
18:PRINT "FORMULA
  S ARE ENTERED"
19:PRINT "LINES 1
  00 THROUGH 120
  0"
20:PRINT "AT YOUR
   OPTION,LABEL"
22:PRINT "TITLE &
   INPUT PARAMET
  ERS"
25:PRINT "ENTER A
   PGM TITLE"
27:INPUT "PROGRAM
   TITLE=";PT$
30:@$(10)="J":@$(
  11)="K":@$(9)=
  "I"
31:@$(12)="L ":@$
  (13)="M ":@$(1
  4)="N ":@$(15)
  ="O ":@$(16)="
  P ":@$(17)="Q
  "
32:@$(18)="R ":@$
  (19)="S ":@$(2
  0)="T ":@$(21)
  ="U ":@$(22)="
  V ":@$(23)="W
  "
33:@$(24)="X ":@$
  (25)="Y ":@$(2
  6)="Z "
35:@$(1)="A ":@$(
  2)="B ":@$(3)=
  "C ":@$(4)="D
  ":@$(5)="E "
37:@$(6)="F ":@$(
  7)="G ":@$(8)=
  "H "
40:REM
41:REM  ENTER
  INPUT PARAMS

42:REM
50:PRINT "PROVIDE
   LBLS FOR<= 18
   PARAMS"
53:PRINT "ENTER T
  HE INPUT PARAM
  S"
54:INPUT "NO.INPU
  T PARAMS=";ZZ
56:IF ZZ>18PRINT
  "ERROR":GOTO 5
  3
57:PRINT "ENTER P
  ARAM LABELS"
58:PRINT "IN THIS
   ORDER:I-Z"
59:WAIT 5
60:FOR II=9TO ZZ+
  8:CLS :FOR I2=
  2TO 22STEP 2:
  CURSOR I2:
  PRINT @$(II):
  NEXT I2:INPUT
  @$(II):NEXT II
61:WAIT 50:GOTO 6
  5
62:"C"SK=1:WAIT 5
  0
65:PRINT "GIVE VA
  LUES TO PARAMS
  "
70:FOR II=9TO ZZ+
  8:PRINT @$(II)
  :INPUT @(II):
  NEXT II
71:REM
72:REM  ENTER AND
   LABEL OUTPUT
  PARAMETERS
73:REM
74:DI$="N":INPUT
  "DOCUMENTATION
  ? (Y/N)";DI$
75:IF SK=1THEN 87
76:PRINT "ENTER O
  UTPUT PARAMS"
77:INPUT "NO.OUTP
  UT PARAMS=";YY
78:IF YY>8PRINT "
  ERROR":GOTO 76
80:PRINT "ENTER O
  UTPUT PARAM LB
  LS A-H"
83:WAIT 5
84:FOR II=1TO YY:
  CLS :FOR I2=2
  TO 22STEP 2:
  CURSOR I2:
  PRINT @$(II):
  NEXT I2:INPUT
  @$(II):NEXT II
85:WAIT 100
87:LPRINT :LPRINT
  PT$:LPRINT LS$
  :LPRINT :IF DI
  $="N"THEN 95
90:FOR II=9TO ZZ+
  8:LPRINT @$(II
  );" = ":LPRINT
  @(II):NEXT II
93:LPRINT
95:REM
96:REM  CARRY OUT
   CALCULATIONS:
  PGM LINES 100-
  1200
97:REM


1203:REM
1204:REM  PRINT
  OUTPUT
1205:REM
1210:LPRINT "COMP
  . RESULTS:":
  LPRINT
1220:FOR II=1TO Y
  Y
1230:LPRINT @$(II
  );" = ":
  LPRINT @(II)
1240:NEXT II
1245:LF 1
1247:REM
1248:REM INSTRUC-
  TIONAL MSG
1249:REM
1250:PRINT "DEF C
   FOR NEW PRO
  B":LPRINT
1272:IF DI$="N"
  THEN 1970
1275:GOTO 1990
1970:FOR II=9TO Z
   Z+8:LPRINT @
  (II)
1980:NEXT II
1990:LF 3
9999:END
```

Listing - GCP (PC-2 without printer)

```
8:"A"WAIT 70:
  CLEAR
9:PRINT "GENERAL
   COMPUTATION P
  GM"
10:REM
11:REM    PROGRAM
   INSTRUCTIONAL
   MESSAGE
12:REM
15:IM$="N":INPUT
   "INSTRUCTION M
   SG?(Y/N)";IM$
16:IF IM$="N"THEN
   30
17:PRINT "BEFORE
   RUNNING, BE SU
   RE"
18:PRINT "FORMULA
   S ARE ENTERED"
19:PRINT "LINES 1
   00 THROUGH 120
   0"
20:PRINT "AT YOUR
    OPTION,LABEL"
22:PRINT "INPUT P
   ARAMETERS"
30:@$(10)="J":@$(
   11)="K":@$(9)=
   "I"
31:@$(12)="L ":@$
   (13)="M ":@$(1
   4)="N ":@$(15)
   ="O ":@$(16)="
   P ":@$(17)="Q
   "
32:@$(18)="R ":@$
   (19)="S ":@$(2
   0)="T ":@$(21)
   ="U ":@$(22)="
   V ":@$(23)="W
   "
33:@$(24)="X ":@$
   (25)="Y ":@$(2
   6)="Z "
35:@$(1)="A ":@$(
   2)="B ":@$(3)=
   "C ":@$(4)="D
   ":@$(5)="E "
37:@$(6)="F ":@$(
   7)="G ":@$(8)=
   "H "
40:REM
41:REM  ENTER
   INPUT PARAMS

42:REM
50:PRINT "PROVIDE
    LBLS FOR<= 18
    PARAMS"
53:PRINT "ENTER T
   HE INPUT PARAM
   S"
54:INPUT "NO.INPU
   T PARAMS=";ZZ
56:IF ZZ>18PRINT
   "ERROR":GOTO 5
   3
57:PRINT "ENTER P
   ARAM LABELS"
58:PRINT "IN THIS
    ORDER:I-Z"
59:WAIT 5
60:FOR II=9TO ZZ+
   8:CLS :FOR I2=
   2TO 22STEP 2:
   CURSOR I2:
   PRINT @$(II):
   NEXT I2:INPUT
   @$(II):NEXT II
61:WAIT 90:GOTO 6
   5
62:"C"SK=1:WAIT 9
   0
65:PRINT "GIVE VA
   LUES TO PARAMS
   "

70:FOR II=9TO ZZ+
   8:PRINT @$(II)
   :INPUT @(II):
   NEXT II
71:REM
72:REM  ENTER AND
    LABEL OUTPUT
   PARAMETERS
73:REM
75:IF SK=1THEN 97
76:PRINT "ENTER O
   UTPUT PARAMS"
77:INPUT "NO.OUTP
   UT PARAMS=";YY
78:IF YY>8PRINT "
   ERROR":GOTO 76
80:PRINT "ENTER O
   UTPUT PARAM LB
   LS A-H"
83:WAIT 5
84:FOR II=1TO YY
85:CLS :FOR I2=2
   TO 22STEP 2:
   CURSOR I2:
   PRINT @$(II):
   NEXT I2:INPUT
   @$(II):NEXT II
90:WAIT 90
95:REM
96:REM  CARRY OUT
    CALCULATIONS:
   PGM LINES 100-
   1200
97:REM
1203:REM
1204:REM  PRINT
     OUTPUT
1205:REM
1210:PRINT "COMP.
      RESULTS:"
1215:PRINT "PRESS
      ENTER TO DI
     SPLAY ANS"
1220:" "FOR II=1
     TO YY:WAIT 8
     0
1230:PRINT @$(II)
     ;" = "
1235:WAIT :PRINT
     @(II)
1240:NEXT II:WAIT
     150:PRINT "R
     EDISPLAY ANS
     WERS?DEF SPA
     CE"
1250:PRINT "PRESS
      DEF C FOR R
     ERUN"
1260:WAIT :PRINT
     "INPUT REVIE
     W?PRESS ENTE
     R"
1270:FOR II=9TO Z
     Z+8:PRINT @(
     II):NEXT II
9999:END
```

Listing - GCP (PC-1 without printer)

```
10:"A"CLEAR
20:BEEP 4:PAUSE
   "*****GCP***
   **":PAUSE "I
   NPUT PARAMS:
   I-T"
25:BEEP 2:INPUT
   "# INPUT PAR
   AMS?(12MAX)"
   ;A(53)
30:I$="I":J$="J
   ":K$="K"
31:L$="L":M$="M
   ":N$="N":O$=
   "O":P$="P":Q
   $="Q":R$="R"
   :S$="S":T$="
   T"
35:FOR A=9TO 8+
   A(53)
40:A$(A+26)=A$(
   A):NEXT A
67:A(54)=0:GOTO
   70
69:"C"PAUSE "GC
   P-RERUN":A(5
   4)=1:FOR A=9
   TO 8+A(53):A
   $(A)=A$(A+26
   ):NEXT A
70:BEEP 2:FOR A
   =9TO A(53)+8
   :PAUSE "INPU
   T":PAUSE A$(
   A);".."; A$(A
   );".."; A$(A)
   ;".."; A$(A):
   INPUT A(A):
   NEXT A
75:IF A(54)=1
   THEN 90
80:BEEP 2:PAUSE
   "AVAIL:8 OUT
   PUT PARA:A-H
   ":BEEP 2:
   INPUT "NO.OF
    OUTPUT PARA
   MS?";A(55)
90:PAUSE "CALCU
   LATION"
501:PAUSE "GCP-R
    ESULTS":
    PRINT "FOR R
    ESULTS-PRESS
     ENTER"
503:FOR A(56)=1
    TO A(55):A(A
    (56)+26)=A(A
    (56)):NEXT A
    (56)
504:A$="A=":B$="
    B=":C$="C=":
    D$="D=":E$="
    E=":F$="F="
507:G$="G=":H$="
    H="
510:" "FOR A(56)
    =1TO A(55):
    PAUSE A$(A(5
    6))
520:PRINT A(A(56
    )+26)
530:NEXT A(56):
    PRINT "REDIS
    PLAY ANS?SHF
    T SPC"
532:PRINT "RERUN
    ?PRESS SHFT
    C"
535:PRINT "INPUT
     REVIEW?PRES
    S ENTER"
540:FOR A(56)=9
    TO A(53)+8:
    PRINT A(A(56
    )):NEXT A(56
    )
```

SAMPLE RUNS (PC-1 and PC-2)

W 14X43 Column, 28 ft. long

```
**STEEL COLUMN**
**    **   **    **

MOM INERT X =
            428
MOM INERT Y =
           45.2
AREA =
           12.6
MEM DEPTH, IN =
          13.66
MEM WIDTH, IN =
          7.995
FLG T-NESS, IN =
           0.53
YIELD STR, KSI =
             36
APPLIED LOAD, K =
             59
KX =
              1
KY =
              1
COL.LGTH.X, FT =
             28
COL.LGTH.Y, FT =
             28

COMP. RESULTS:

AXIAL STRESS, KSI =

    4.682539683
ALLOW AX STRESS, =

    4.745046668
```

```
GEN COM PGM
  RESULTS
A=
   4.682539683
B=
   4.745046668

          428.
          45.2
          12.6
         13.66
         7.995
          0.53
           36.
           59.
            1.
            1.
           28.
           28.
```